


Lesson 34: A 90-degree eyepiece

In this lesson we will design an eyepiece. To show how simple this kind of job is on modern lens design software, we will let the computer do almost all the work. (We *like* making the computer do all the work!) The result will be an 8-element lens with good performance.

Let us assume that the image formed by the telescope objective is 1.2 inches in diameter, and we want the viewer to see an undistorted field of view of 90 degrees. This kind of lens can be designed in two ways: from the objective to the eye, or the reverse. In this example we will select the former. (In lesson 37 we will do it the other way around.)

We will aim light from a distant objective into the lens, initially at a point one inch inside the lens, and require that the light exit at 45 degrees from the axis.



Here is the input for a run on DSEARCH, which will find configurations for us.

```
CORE 16          ! if you have a multicore PC, by all means use it
PROJECT         ! to see how long the process took

DSEARCH 1 QUIET
SYSTEM
ID DSEARCH SAMPLE
OBA -1 .6 .05    ! object starts 1 inch inside the lens
WAVL 0.6563 0.5876 0.4861 ! visible light

AFOCAL          ! output will be collimated
UNITS INCH      ! lens is in inches
OSNA 0.08333    ! design for an F/6 objective lens
END

GOALS
ELEMENTS 8
BACK 0 0        ! control these with limits in the special section
TOTL 0 0
STOP TELECENTRIC ! the input beam comes from a distant objective
RT 0.5
FOV 0.0 .3 .6 .9 1 ! correct three field points
FWT 5.0 3.0 3 3 3 ! with these weights
ANNEAL 200 20 Q  ! annealing is slower, but works better
SNAPSHOT 10      ! watch the progress
QUICK 30 40      ! do a quick scan of all designs, 20 passes. Then 20 more with
END              ! real rays

SPECIAL PANT    ! vary the object position within the eyepiece
VY 0 TH0
END
```


```

SPECIAL AANT
ADT 1 .1 1
AEC .15 1 1 ! require edge thicknesses of 0.15 inches or more
ACA 60 1 1 ! stay away from critical angle refraction
M .7 1 A BACK ! specify 0.7 inch eye relief
LUL 6 1 1 A TOTL ! allow lens length up to 6 inches
M -1 10 A P HH 1 ! require full-field ray angle of 45 degrees
M 0 1 A P YA 1 0 0 0 LB1 ! correct for pupil aberration at two fields
M 0 1 A P YA .5 0 0 0 LB1 ! "LB1" means last surface but 1, or the eye point
M 0 1 A P YA 1 ! control distortion this way
S GIHT
END


GO ! run the search
PROJECT ! see how long it took

```

Here is the drawing that comes back from DSEARCH.



These are all plausible eyepiece configurations. The best one is at the top, saved as DSEARCH08.RLE in this case, and it opens automatically in PAD. It's not bad, for starters.




The program has created an optimization MACro and loaded it into an editor window. Here you see the targets the program has generated along with the special targets given in the DSEARCH input.

```

PANT
  VY 0 TH0
  VLIST RD ALL
  VLIST TH ALL
  VY 1 GLM
  VY 3 GLM
  VY 5 GLM
  VY 7 GLM
  VY 9 GLM
  VY 11 GLM
  VY 13 GLM
  VY 15 GLM
END
AANT P
AEC
ACC
GSR 0.500000 5.000000 4 M 0.000000
GNR 0.500000 3.000000 4 M 0.300000
GNR 0.500000 3.000000 4 M 0.600000
GNR 0.500000 3.000000 4 M 0.900000
GNR 0.500000 3.000000 4 M 1.000000
ADT 6 .1 1
AEC .15 1 1 ! REQUIRE EDGE THICKNESSES OF 0.15 INCHES OR MORE
ACA 60 1 1 ! STAY AWAY FROM CRITICAL ANGLE REFRACTION
M .7 1 A BACK ! SPECIFY 0.7 INCH EYE RELIEF
LUL 6 1 1 A TOTL ! ALLOW LENS LENGTH UP TO 6 INCHES
M -1 10 A P HH 1 ! REQUIRE FULL-FIELD RAY ANGLE OF 45 DEGREES
M 0 1 A P YA 1 0 0 0 LB1 ! CORRECT FOR PUPIL ABERRATION AT TWO FIELDS
M 0 1 A P YA .5 0 0 0 LB1 ! LB1 MEANS LAST SURFACE BUT 1, OR THE EYE POINT
M 0 1 A P YA 1 ! CONTROL DISTORTION THIS WAY
S GIHT
END

```

Let's run this and watch the lens improve. Here is what we get after some optimization and annealing:




We'd like to do somewhat better. Assigning OPD fans to PAD Display number 2, we see errors on the order of one wave at full field.

What happens if we add some OPD targets to the merit function? In the AANT file add

```
GNO 0 .01 4 M 0
GNO 0 .01 4 M .3
GNO 0 .01 4 M .6
GNO 0 .01 4 M .9
GNO 0 .01 4 M 1
```


After more optimization and annealing, the lens is better:



Is this good enough? Let's check some things. We want to avoid the notorious "kidney-bean" effect, which shows up if there is much spherical aberration of the pupil. As your eye moves about, parts of the field disappear here and there.

It's easy to check. Zoom in on the eye position and run the Pad Scan™ by clicking the button . As hoped, the rays stay nicely aimed at the eye point. Not perfectly, to be sure – but then, lens design is all about tradeoffs, isn't it? We judge the small errors to be quite good enough.

Next we have to examine distortion. Some poor-quality eyepieces show significant distortion, and we have to check. The command **GDIS 21 G** produces this picture:




Wow! That's almost perfect! We got that fine performance by including an item in the merit function that takes the full-field Y coordinate and subtracts the GIHT. But how can that work? The units are all wrong!

Not so. This lens is in AFOCAL mode, and the output Y-coordinate is actually the ray angle from the axis, in radians. The GIHT is similarly the angle in radians found paraxially. If the two match, presto! No distortion.


Now the hard part. We have to check the image quality. It's about 1/2 wave of lateral color, and this may or may not be too much. To analyze this, we will assign 10 wavelengths to the lens, weighted according to the spectrum of an astronomical object and the human eye. First, we delete the curvature solve, so the radius won't change as the spectrum does.

```
CHG
NOP ! delete all pickups and solves
END
MSW
```


When the Spectrum Wizard opens, we select Astronomical sources and Sun, moon, planets. Then we click Visual and select Visual, bright light. Click the selection for 10 wavelengths, and then Get Spectrum.




Here is the spectrum for this combination of source and detector. Click the Fine setting, and shift the spectrum a little to the right. Then click Apply to lens and Close.




Close the Wizard and open the dialog **MPF**. This will show the diffraction point-spread over the field. Select Show visual appearance, Magnify by 4, and execute.



Hmmm.... The image at about 0.9 field is not as good as at full field. Let's go back to the 3-color version, increase the OPD weight there to 0.02, and reoptimize. Yes, definitely worth doing!





While this is not perfect, it is certainly diffraction limited over most of the field. In practice, few telescopes are this good, so one must judge whether it is worthwhile to aim higher. The blur shown here is comparable to the resolution of your eye, so the image will probably look as sharp as you can see.

This lens was the top one on the DSEARCH list, but we have sometimes found a better lens by trying the others. Feel free to examine them all -- and be aware that, since we used the simulated annealing feature in this lesson, the results that come back will be somewhat different each time. So by all means run DSEARCH several times, optimizing and examining the results each time. And experiment with the value of RSTART too. (This example used the default, but a value of from 5 to 10, which is much shorter, will give another set of possibilities.)

If we were happy with this eyepiece, the next step would be to change to real glasses. Read about the IRG and ARGLASS features. They make this part of the job easy to do. It would also be wise to replace the telecentric object with a real OBA object formed by an entrance pupil at the anticipated distance from the eyepiece. That will move the eye point out slightly, and you might get some benefit from then letting it come back in again.